

```

//*****
//*** INSERT VARIABLE MACRO
//*** HOTDOCS DOCUMENT ASSEMBLER V3.0
//*** CAPSOFT DEVELOPMENT CORPORATION
//*****

revealOn := ?RevealCodesActive
//*** CHECK FOR EXISTING FIELD

RUN("hdchkfld.wcm")

Display(On!)

//*** SET BLOCK ACTIVE AND SELECTSTR
blockActive := ?BlockActive
spcCnt := 0
IF (blockActive = 0)
    SelectStr := ""
ELSE
    SelectStr := ?SelectedText
    len := STRLEN(SelectStr)
    REPEAT
        spcStr := SUBSTR(SelectStr;len;1)
        IF (spcStr = " ")
            spcCnt := spcCnt + 1
            len = len - 1
            SelectStr := SUBSTR(SelectStr;1;len)
        ENDIF
    UNTIL ((spcStr <> " ") OR (len = 0))
ENDIF

//*** LOAD AND CALL THE DLL
OrgFile := ?Path+?Name
Result := 0
IF (?MajorVersion <= 6)
    DllLoad(HotdocsDll;"HD3.dll")
    wpid = 32 * 4
ELSE
    DllLoad(HotdocsDll;"HD3_32.dll")
    wpid = 32 * 8
ENDIF
Display(On!)
DllCall(
    HotdocsDll;
    "InsertVar";
    RetStr:AnsiString;
    {OrgFile;1+wpid;ADDRESS(Result);SelectStr}

```

```

)
Display(Off!)

/** IF SOMETHING IS WRONG, QUIT
IF (Result=0)
    DllFree(HotdocsDll)
    IF (revealOn)
        RevealCodes(On!)
    ENDIF
    Quit
ENDIF
/** DELETE SELECTED TEXT
IF (blockActive <> 0)
    Type("x")
    DeleteCharPrevious()
ENDIF

/** INSERT FIELD
CALL(InsField@)
FOR (i;1;i<=spcCnt;i+1)
    Type(" ")
ENDFOR

/** IF REPLACE ONCE, QUIT
IF (Result = 1)
    IF (revealOn)
        RevealCodes(On!)
    ENDIF
    Quit
ENDIF

/** REPLACE ALL OR CONFIRM

/** SETUP SEARCH
capsBM := "CapsBM"
BookmarkCreate(capsBM)
capsHTBM := "CapsHTBM"

ONNOTFOUND(Finish@)
TResult := 1
SearchCaseSensitive(State:No!)
SearchFindWholeWordsOnly(State:No!)
SearchInSelection(State:No!)
SearchWrap(State:No!)
MatchLimit(State:No!;0)
MatchSelection()

```

```

MatchWithAttributes(State:No!)
MatchWithFont(State:No!)
MatchWithFontSize(State:No!)
/** LOOP THROUGH ALL INSTANCES OF SELECTED TEXT
REPEAT
  SearchString(SelectStr)
  SearchNext(Extended!)
  IF (?TextColorRed = 1 AND ?TextColorGreen = 2 AND ?
TextColorBlue = 254)
    GO(RepeatEnd@)
  ENDIF
  IF (Result=3)
    /** GET FOUND TEXT
    TSelectStr := ?SelectedText
    TResult := 0
    Display(On!)
    DllCall(
      HotdocsDll;
      "InsertVar";
      RetStr: AnsiString;
      {OrgFile;4+wpid;ADDRESS(TRResult);TSelectStr}
    )
    Display(Off!)
  ENDIF

  /** RESULT WAS ZERO, CANCEL WAS PRESSED
  IF (TRResult = 0)
    GO(Finish@)
  ENDIF

  /** RESULT WAS THREE, YES TO ALL, IE STOP CONFIRMING
  IF (TRResult = 3)
    TResult := 1
    Result := 2
  ENDIF

  /** RESULT WAS ONE, REPLACE THIS INSTANCE
  IF (TRResult = 1)
    CALL(InsField@)
  ENDIF
  LABEL(RepeatEnd@)
UNTIL (Result = 0)

/** CLEANUP STUFF
Label(Finish@)
SelectMode(Off!)
BookmarkFind(capsBM)

```

```
BookmarkDelete (capsBM)
IF (revealOn)
    RevealCodes (On!)
ENDIF
DllFree (HotDocsDll)
QUIT
```

```
LABEL (InsField@)
    Type ("«»")
    PosCharPrevious ()
    Type ("X")
    SelectCharPrevious
    TextColor (;1;2;254)
    PosSelectBottom ()
    SelectMode (Off!)
    DeleteCharPrevious ()
    IF (qcOn) QuickCorrect (Off!) ENDIF
    IF (cflOn) CapBeginSentence (Off!) ENDIF
    IF (ssqOn) SingleSmartQuote (Off!;;) ENDIF
    IF (dsqOn) DoubleSmartQuote (Off!;;) ENDIF
    IF (ticOn) IrregularCap (Off!) ENDIF
    Type (RetStr)
    IF (qcOn) QuickCorrect (On!) ENDIF
    IF (cflOn) CapBeginSentence (On!) ENDIF
    IF (ssqOn) SingleSmartQuote (On!;;) ENDIF
    IF (dsqOn) DoubleSmartQuote (On!;;) ENDIF
    IF (ticOn) IrregularCap (On!) ENDIF

    PosCharNext ()
RETURN
```